Pictures and movies from singularity theory

Sergei Duzhin

duzhin@u-aizu.ac.jp

0. Introduction

0.1 In this lesson, we will talk about singularities of functions, curves and surfaces and how the capabilities of Mathematica (symbolic, numerical and graphical) can be used in the study of singular phenomena.

0.2 This worksheet, like any Mathematica worksheet, consists of explanatory text , input and output. You can: (1) read the explanatory text, (2) execute the commands written in the input lines (to do so, place the cursor on that line and press Shift-Return) and (3) look at the results contained in the output.

0.3 In Mathematica, there is a special type of output called "animated graphics". Let us have a drill in using it.

Place the cursor anywhere in the picture below and double-click the mouse’s left button. The curve will begin moving. You can control the velocity and the direction of motion using the VCR-like buttons below, in the margin of the window.

Try it now! To stop the animation, press the mouse once anywhere in the picture area.

0.4 I will explain the mathematical meaning of this picture below (see the section "Plane Curves and their singularities").

0.5 Here is an example of a 3D animation:

1. Critical points of functions in one variable

1.1 As the simplest example, consider a polynomial of second degree given by the formula

f[x_]= x^2 + b x + c
           2
c + b x + x

If we assign certain values to the coefficients, for example,

b = 2; c = -3;

then the polynomial becomes

f[x]
            2
-3 + 2 x + x

Let us draw the graph of this function:

Plot[f[x],{x,-4,2}]

In this simple graph, you can see the first example of a singularity.

1.2 Of all the points in x axis, which point is most different from all other points? To answer this question, let us look at the behaviour of the curve near a given point. If you look at the graph in the vicinity of any point to the left of -1 and imagine that you are travelling along the x axis from left to right -- you will see that the curve goes down (the functiondecreases ). If you look at any point to the right of the point -1, you will see that the curve goes up, i.e. the function increases , in a whole neighbourhood of the chosen point.

1.3 There is only one point, x=-1 , where the behaviour of the function changes: it it is decreasing to the left of this point and increasing to the right of this point. The point x=-1 is said to be a critical point of the function f[x]=x^2+2x-3 . Using another terminology, this is a point where a catastrophe occurs. Singularity theory is a branch of mathematics that studies critical points of functions and singular points of manifolds.

1.4 In terms of differential calculus, a critical point of a function of 1 variable is a point where its derivative becomes 0. Let us compute the derivative and find the value for which it becomes 0:

f1[x_] = D[f[x],x]

2 + 2 x

Solve[f1[x]==0,x]

{{x -> -1}}

The found value ofx is the value that we used above!

1.5 Exercise. Assign some values to the coefficients b and c:

b =  ; c = plot the graph of the polynomial:

f[x_] = x^2 + b x + c
Plot[f[x],{x,-5,5}]

find the singular point and describe the behaviour of the graph to the left and to the right of that point. To check the result by Mathematica, use the following lines of input:

f1[x] = D[f[x],x]
Solve[f1[x]==0,x]

1.6 A more interesting example is given by the formula

f[x_] = x^3 - 3x^2 + 2x
         2    3
2 x - 3 x  + x

Looking at the graph of this function:

Plot[f[x],{x,-1,3}]

you can see two critical points: a maximum near 1.5 and a minimum near 2.5. The following command finds the numeric values of these points to 5 digits:

NSolve[D[f[x],x]==0,x]
{{x -> 0.42265}, {x -> 1.57735}}

1.7 A polynomial of degree n may have up to n-1 critical points. Here is the so called Chebyshev polynomial of degree 9:

ChebyshevT[9,x]

           3        5        7        9
9 x - 120 x  + 432 x  - 576 x  + 256 x

Plot[{ChebyshevT[9,x],1,-1},{x,-1.2,1.2}]

A remarkable feature of Chebyshev polynomials is that they have the maximal possible number of critical points, and all the values of the function at these points (maxima and minima) lie on only 2 horizontal lines.

2. Animation: families of functions

2.1 Suppose we have a family of functions, i.e. a function depending on a parameter. When the parameter changes, the critical points of the function can move, merge, split, appear and disappear.

2.2 In the following example, we plot the function x^3+ax where x is a variable ranging between -2 and 2, and a is a parameter that takes values from 1 to -2. For every value of a taken with the step 0.1 (e.g. a=1, a=0.9, 0.8 ...) we plot the graph of the functionx^3+ax and then put these pictures together to make a movie.

Do[Plot[ x^3+a*x, {x, -2, 2},
PlotRange -> {{-2,2}, {-8,8}}],
{a, 1, -2, -1/5}]

In this example, the initial function x^3 +x has no critical points; when a becomes 0, we see that a critical point (inflexion point, in this case) appears, then this point splits into two different simple singularities (a maximum and a minimum).

2.3 Here is one more similar example, where three critical points merge into one:

Do[Plot[ x^4+a*x^2, {x, -2, 2},
PlotRange -> {{-2,2}, {-6,8}}],
{a,-4,1,1/4}]

2.4 And in the following animation the graph only moves in the horizontal direction. Its critical points move with the graph.

Do[Plot[Sin[x-t],{x,0,4 Pi},
  PlotRange -> {{0, 4 Pi}, {-2, 2}}],
  {t, Pi/6, 2 Pi, Pi/6}]

2.5 Exercise . Try to change the previous command so that the waves start going in the opposite direction (from right to left). Hint: Only one symbol has to be replaced.

3. Plane curves and their families

3.1 So far, we have discussed critical of functions. In this section, we will look at plane curves and their singular points.

A point on the curve is called regular, if the curve is smooth in the vicinity of that point. A point is singular, if the curve is not smooth. The most typical singularities of plane curves are self-intersections and cusps .

3.2 Example: A curve with all regular points.

ParametricPlot[{t^2+1,t*(t^2+1)},{t,-3,3},
PlotRange -> {{-3,3},{-3,3}}]

3.3 Example: A curve with one cusp.

ParametricPlot[{t^2,t^3},{t,-3,3},
PlotRange -> {{-3,3},{-3,3}}]

3.4 Example: A curve with one self-intersection.

ParametricPlot[{t^2-1,t*(t^2-1)},{t,-3,3},
PlotRange -> {{-3,3},{-3,3}}]

3.5 These three examples belong to one family of curves given by parametric equations x = t^2-a, y = t(t^2-a) . Example 3.2 corresponds to a=-1 , Example 3.3 - to a=0, and Example 3.4 - to a=1. The following command generates a movie which includes these three curves. Run the animation by double-clicking the picture.

Do[ParametricPlot[{t^2-a,t(t^2-a)},{t,-3,3},
PlotRange -> {{-3,3},{-3,3}}],{a,-1,2,1/4}]

3.6 Here is a plane curve with 4 self-intersections:

ParametricPlot[{2 Cos[t]+Cos[5 t],2 Sin[t]+Sin[5 t]},
{t,0,2 Pi},AspectRatio -> 1]

3.7 And here is an animation which includes the previous curve as a stillshot:

Do[
ParametricPlot[{2 Cos[t]+a*Cos[5 t],2 Sin[t]+a*Sin[5 t]},
{t,0,2 Pi},PlotRange -> {{-3,3},{-3,3}},
AspectRatio -> 1], {a,-1,1,0.1}]

3.8 Exercise . Insert any formulas containing t in the following line of input and press Shift- Return:

x = ; y = ;

Choose the initial and the final values of t:

t0 = ; t1 = ;

Then run the plotting command:

ParametricPlot[{x,y},{t,t0,t1}]

What are the singularities of the curve thus obtained?

4. Functions in two variables

4.1 For the functions of one variable, singularity theory is very simple: a typical function has several maxima and minima that are arranged along the x axis in an interchanging way. In the case of functions of two variables, there are 3 types of stable critical points: maximum, minimum and saddle point, and they can be arranged in the plane following various patterns. 4.2 To have a better look of the 3D pictures, we will use the procedure "SpinShow" from the "Animation" package, so we begin by loading it (be sure to press Shift-Enter in the next line right now!):

<<"Graphics‘Animation‘"

Now execute the following commands:

f[x_,y_] = (x^2 + 3 y^2) Exp[1 - x^2 - y^2]

      2    2
 1 - x  - y    2      2
E            (x  + 3 y )

surf = Plot3D[f[x,y],{x,-2,2},{y,-2.5,2.5},
Boxed -> False, AxesEdge -> None,
PlotPoints -> 21, PlotRange -> {0,3.1}]

4.3 You see a surface in 3 dimensional space. You can have a look at this surface from any point in the space.

The point of view is specified by 3 numbers that are entered as the ViewPoint option in the Plot3D command. Run the following command; then replace the numbers 2, -0.8, 0.7 by any numbers you like and run it again.

Plot3D[f[x,y],{x,-2,2},{y,-2.5,2.5},
Boxed -> False, AxesEdge -> None,
PlotPoints -> 21, ViewPoint -> {2,-0.8,0.7}]

4.4 The next command will let you see the surface in motion. After all the stillshots are ready, choose "Closed Group" from the "Cell" menu.

SpinShow[surf,SpinRange->{0 Degree, 180 Degree},Frames->12]

4.5 Now we will discuss the critical points of the function f[x,y]. These correspond to the points of the surface where the tangent plane is horizontal. In our example, there are 5 such points: 2 points of maximum (like mountain summits), 1 point of minimum (like a pit) and 2 saddle points (like mountain passes).

4.6 A mountain (topographical) map can be produced with the comm and ContourPlot:

ContourPlot[f[x, y],{x,-2,2},{y,-2.5,2.5},
ContourShading -> True,
ColorFunction -> Hue,
PlotPoints -> 20,
Contours -> 13]

4.7 It is also very enlightening to see the horizontal level sections of the surfaces. The following command produces an animation where the surface is cut by the plane z=h, where h increases from 0 to 3 with the step 0.5:

Do[Show[{surf,Plot3D[h,{x,-2,2},{y,-2,2},
Boxed -> False, AxesEdge -> None,
BoxRatios -> {2,2,1},PlotRange -> {0,3.1},
DisplayFunction -> Identity]}],{h,0,3,0.5}]

4.8 Using differential calculus, critical points of a function in two variables can be found as solutions to the system of equations "both partial derivatives are zero":

eqn1 = D[f[x,y],x]==0
eqn2 = D[f[x,y],y]==0

        2    2             2    2
   1 - x  - y         1 - x  - y      2      2
2 E            x - 2 E            x (x  + 3 y ) == 0

        2    2             2    2
   1 - x  - y         1 - x  - y      2      2
6 E            y - 2 E            y (x  + 3 y ) == 0

Solve[{eqn1,eqn2},{x,y}]

{{x -> -1, y -> 0}, {x -> 0, y -> 0}, {x -> 1, y -> 0},
  {y -> -1, x -> 0}, {y -> 1, x -> 0}}

We found 5 solutions which agrees with the picture of the surface above.

4.9 To understand whether a critical point is a maximum, a minimum or a saddle point, we find the second derivatives and write them as a matrix 2x2:

dxx = D[f[x,y],x,x] /. {x->-1,y->0};
dxy = D[f[x,y],x,y] /. {x->-1,y->0};
dyy = D[f[x,y],y,y] /. {x->-1,y->0};
MatrixForm[{{dxx,dxy},{dxy,dyy}}]

-4   0
0    4

Since the determinant of this matrix is negative ( det<0 ), the point (-1,0) is a saddle point. If the determinant is positive ( det>0 ), then we have a minimum in the case dxx>0 and a maximum in the case dxx<0.

4.10 Exercise . Repeat the calculation of the matrix for other critical points.

4.11 Exercise . Repeat all the previous commands for the following function:

f[x_,y_] = Cos[x]+Cos[y]+Cos[x+y]

s = Plot3D[f[x,y],
{x,0,2 Pi},{y,0,2 Pi}, Boxed -> False,
AxesEdge -> None]

SpinShow[s]

5. Singular surfaces

5.1 In this section, we will show some pictures of singular surfaces.

A point of the surface is called regular, if in the small neighbourhood of this point the surface is smooth, i.e. looks like a piece of plane. Otherwise the point is said to be singular.

The vertex of the conical surface provides an example of the simplest isolated singularity:

cone=ParametricPlot3D[{u Cos[v],u Sin[v],u},
{u,-1,1},{v,0,2 Pi},Axes->False, Boxed->False]

5.2 The points that lie on the intersection line of two planes give the simplest example of non-isolated singularities:

p1 = Plot3D[-x,{x,-1,1},{y,-1,1},
DisplayFunction->Identity]
p2 = Plot3D[y,{x,-1,1},{y,-1,1},
DisplayFunction->Identity]
Show[{p1,p2},AspectRatio->1,
DisplayFunction->$DisplayFunction,
Axes->False,Boxed->False]

Here is a curvilinear surface which has the same type of singular points.

ParametricPlot3D[{u(u^2-1),v,u^2},
{u,-1.3,1.3},{v,-1,1},Axes->False, Boxed->False]

5.3 The next surface (called Enneper’s surface ) looks more complicated, but its singularities are quite simple: they belong to the same type "intersection of two planes".

ParametricPlot3D[{Re[E^(I*th)*r - (E^(3*I*th)*r^3)/3],
Re[I*E^(I*th)*r + I/3*E^(3*I*th)*r^3],
Re[E^(2*I*th)*r^2]}, {th, 0, 2 Pi}, {r, 0, 3},
ViewPoint -> {2.8, -1.9, 0.1}, Boxed->False,
PlotPoints -> {60, 20}, Axes -> False, PlotRange -> All,
LightSources ->{{{0.7,0,0.7}, RGBColor[0.9,0,0]},
{{0.6,0.6,0.6}, RGBColor[0,0.9,0]},
{{0,0.7,0.7}, RGBColor[0,0,1]}}]

5.4 Our last example is a surface with 3 simple isolated singularities which is constructed using Chebyshev polynomials (see section 1). It is called "Chmutov’s surface of degree 3".